*Replication file 01c_Onset_limited_sample_EU
*Article: Counterfactual Coercion: Could harsher sanctions against Russia have prevented the worst?
*Authors: Thies Niemeier, Gerald Schneider


***************************************************************
***EU***
***************************************************************

set seed 1234

*Prepare data
use "Dataset.dta", clear
keep if sender=="EU"

gen ln_oil_gas_value_2014 = ln(oil_gas_value_2014+1)
gen sender_colony=0
replace sender_colony=1 if ht_colonial==2 | ht_colonial==3 | ht_colonial==6 | ht_colonial==7 | ht_colonial==8 | ht_colonial==9 | ht_colonial==10

gen sender_additional=cond(threatUS==1 | impositionUS == 1, 1, 0)
gen only_threat=cond(threatEU==1 & impositionEU == 0, 1, 0)

gen sender_trade = ln_EU_Trade_Eurostat
gen coup_dummy = coup1
replace coup_dummy = 0 if coup_dummy == 1
replace coup_dummy = 1 if coup_dummy == 2

* Dependent variable : 1 if a threat or sanction case started in the dyad
xtset ccodecow year
replace caseid=0 if caseid==.
gen sanctiononset = (caseid-l.caseid)
replace sanctiononset=1 if sanctiononset > 1 & !missing(sanctiononset)
replace sanctiononset=0 if sanctiononset < 0
replace sanctiononset=. if sanctiononset == 0 & (sanction_dyad == 1 | threat_dyad == 1)
tab sanctiononset
gen sanction_test = sanctiononset if year > 2009

* lag time-series variables
sort ccodecow year
by ccodecow: gen l_v2x_polyarchy = v2x_polyarchy[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_gd_ptss = gd_ptss[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_coup_dummy = coup_dummy[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_one_sided_violence = one_sided_violence[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_conflict = conflict[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_mid_terr_integrity = mid_terr_integrity[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_ln_GDPpc_imputed = ln_GDPpc_imputed[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_sender_trade = sender_trade[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_ln_oil_gas_value = ln_oil_gas_value_2014[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_defense_alliance = defense_alliance[_n-1] if year==year[_n-1]+1

* create dummy variables
tabulate l_gd_ptss, generate (pol_terr)


** Filter for cases of importance
keep if pot_sanctioned_countries == 1 & !missing(sanctiononset)

* sortieren nach Jahr, zur Vorbereitung RF model
gen u=0
replace u=1 if year >= 2009
sort u

** Imposition
* Random Forest Model
rforest sanctiononset l_v2x_polyarchy pol_terr* l_coup_dummy ///
l_one_sided_violence l_conflict l_mid_terr_integrity ///
l_ln_GDPpc_imputed l_sender_trade l_ln_oil_gas_value ///
sender_colony l_defense_alliance in 1/2364, type(class) iter(1500) numvars(15)

*Variable Importance
ereturn list
matrix list e(importance)
* write Variable importance to excel file
putexcel set "Supplemental_Material\Variable_Importance\Variable_Importance_Onset_EU_RF.xlsx", sheet("M") replace
putexcel A1=matrix(e(importance)), names

* Evaluation of model performance
** predictions
predict randonsEU
predict randonsEU0 randonsEU1, pr

* Confusion Matrix
** Sensitivity 100, Specificity 91.4
tab2xl sanction_test randonsEU using Supplemental_Material\Prediction_Output\Confusion_Matrixes\EU_Onset_RF, row(1) col(1)
diagtest sanction_test randonsEU

* AUPR .10
prtab sanction_test randonsEU1, l2title("RF Onset", box bexpand) 
graph save "Graph" "Supplemental_Material\Prediction_Output\Roc-curves\AUPR_Onset_RF_EU.gph", replace

* .21
kap sanction_test randonsEU
